

Mini Java Grammar

Leela Pakanati, Manoj Kurapati

Program ::= MainClassDecl ClassDeclList

MainClassDecl ::= class ID { public static void main (String[] ID) { StmtList } }

ClassDeclList = ClassDecl ClassDeclList
| ϵ

ClassDecl ::= class ID [extends ID] { ClassVarDeclList MethodDeclList }

ClassVarDeclList = ClassVarDecl ClassVarDeclList
| ϵ

ClassVarDecl ::= Type ID;

MethodDeclList = MethodDecl MethodDeclList
| ϵ

MethodDecl ::= public Type ID ([Formal FormalList]) { StmtList return Expr; }

FormalList = , Formal FormalList
| ϵ

Formal ::= Type ID

Type ::= int | boolean | ID

StmtList = Stmt StmtList
| ϵ

Stmt ::= if (Expr) BracketStmt else BracketStmt
| while (Expr) BracketStmt
| AtomicStmt

BracketStmt ::= { AtomicStmtList }
| AtomicStmt

AtomicStmtList = AtomicStmt AtomicStmtList
| ϵ

AtomicStmt ::= Type ID = Expr;
| System.out.println(Expr);
| ID = Expr;

ExprList = , Expr ExprList
| ϵ

Expr ::= OrOperand Expr'

Expr' ::= || OrOperand Expr'
| ϵ

OrOperand ::= AndOperand OrOperand'

OrOperand' ::= && AndOperand OrOperand'
| ϵ

AndOperand ::= EqualityOperand AndOperand'

AndOperand' ::= == EqualityOperand AndOperand'
| != EqualityOperand AndOperand'
| ϵ

EqualityOperand ::= RelOperand EqualityOperand'

EqualityOperand' ::= < RelOperand EqualityOperand'
| <= RelOperand EqualityOperand'
| >= RelOperand EqualityOperand'
| > RelOperand EqualityOperand'
| ϵ

RelOperand ::= AddOperand RelOperand'

RelOperand' ::= + AddOperand RelOperand'
| - AddOperand RelOperand'
| ϵ

AddOperand ::= MulOperand AddOperand'

AddOperand' ::= * MulOperand AddOperand'
| / MulOperand AddOperand'
| ϵ

MulOperand ::= ! UnaryOperand
| - UnaryOperand
| UnaryOperand

UnaryOperand ::= (ParenOperand)
| ParenOperand

ParenOperand ::= Value . ID ([Expr ExprList])
| Value

```
Value ::= new ID ( )  
       | ID  
       | this  
       | Integer  
       | null  
       | true  
       | false
```